Minimization example

```
> restart:
  with(LinearAlgebra):
```

Define the function to minimize

```
> f := x^2-x*y ;
```

$$f := x^2 - x\,y \quad (1)$$

Define the unilateral constraints

```
> g1 := 1 - x^2 - y^2 ;
```

$$g1 := 1 - x^2 - y^2 \quad (2)$$

```
> g2 := 1-(x-1)^2 - y^2 ;
```

$$g2 := 1 - (x-1)^2 - y^2 \quad (3)$$

Plot the boundary of the constraints

```
> plot( [[cos(t),sin(t),t=0..2*Pi],[1+cos(t),sin(t),t=0..2*Pi]],
  scaling=CONSTRAINED) ;
```



Bouild the Lagrangian

```
> L := f - mu1 * g1 - mu2 * g2 ;
```

$$L := x^2 - x\,y - \mu 1\left(1 - x^2 - y^2\right) - \mu 2\left(1 - (x-1)^2 - y^2\right) \quad \textbf{(4)}$$

First order KKT conditions

```
> EQ1 := diff( L, x ) ;
  EQ2 := diff( L, y ) ;
  EQ3 := mu1*g1 ;

  EQ4 := mu2*g2 ;
```

$$EQ1 := 2x - y + 2\mu 1\,x - \mu 2\,(-2x + 2)$$
$$EQ2 := -x + 2\mu 1\,y + 2\mu 2\,y$$
$$EQ3 := \mu 1\left(1 - x^2 - y^2\right)$$
$$EQ4 := \mu 2\left(1 - (x-1)^2 - y^2\right) \quad \textbf{(5)}$$

```
> Hess := <<diff(L,x,x),diff(L,x,y)>|<diff(L,y,x),diff(L,y,y)>> ;
```

$$Hess := \begin{bmatrix} 2 + 2\mu 1 + 2\mu 2 & -1 \\ -1 & 2\mu 1 + 2\mu 2 \end{bmatrix} \quad \textbf{(6)}$$

Try mu1=0 and mu2 = 0

```
> EQS00 := subs(mu1=0,mu2=0,[EQ||(1..4)] ) ;
```

$$EQS00 := [2x - y, -x, 0, 0] \quad \textbf{(1.1)}$$

```
> SOL00 := solve( EQS00, {x,y} ) ;
```

$$SOL00 := \{x = 0, y = 0\} \quad \textbf{(1.2)}$$

Check that the solution satisfy the constraints

```
> subs( SOL00, [g1,g2] ) ;
```

$$[1, 0] \quad \textbf{(1.3)}$$

Ok, it is feasible, now check second order condition.
The active constraint is only g2 ==> find the kernel of the gradient of g2

```
> gradg2 := subs(SOL00,<diff(g2,x)|diff(g2,y)>) ;
```

$$gradg2 := \begin{bmatrix} 2 & 0 \end{bmatrix} \quad \textbf{(1.4)}$$

Find the null space of gradg2

```
>  N := op(NullSpace(gradg2)) ;
```

$$N := \begin{bmatrix} 0 \\ 1 \end{bmatrix} \quad \textbf{(1.5)}$$

Check if the hessian is SPD in the kernel

```
> Hess00 := subs(SOL00,Hess) ;
```

$$Hess00 := \begin{bmatrix} 2 + 2\mu 1 + 2\mu 2 & -1 \\ -1 & 2\mu 1 + 2\mu 2 \end{bmatrix} \quad \textbf{(1.6)}$$

```
> Transpose(N).Hess00.N ; subs(mu1=0,mu2=0,%) ;
```

$$2\mu 1 + 2\mu 2$$
$$0 \quad \textbf{(1.7)}$$

Try mu1=0 and mu2 free

```
> EQS00 := subs(mu1=0,[EQ||(1..4)] ) ;
```

$$EQS00 := \left[2\,x - y - \mu 2\,(-2\,x + 2),\ -x + 2\,\mu 2\,y,\ 0,\ \mu 2\left(1 - (x-1)^2 - y^2\right)\right] \quad \textbf{(2.1)}$$

```
> SOLXY := solve( EQS00[1..2], {x,y} ) ;
```

$$SOLXY := \left\{x = \frac{4\,\mu 2^2}{4\,\mu 2 - 1 + 4\,\mu 2^2},\ y = \frac{2\,\mu 2}{4\,\mu 2 - 1 + 4\,\mu 2^2}\right\} \quad \textbf{(2.2)}$$

```
> subs(SOLXY,EQS00[4]/mu2) ; EQmu2 := simplify(%) ;
```

$$1 - \left(\frac{4\,\mu 2^2}{4\,\mu 2 - 1 + 4\,\mu 2^2} - 1\right)^2 - \frac{4\,\mu 2^2}{\left(4\,\mu 2 - 1 + 4\,\mu 2^2\right)^2}$$

$$EQmu2 := \frac{4\,\mu 2^2\left(4\,\mu 2^2 + 8\,\mu 2 - 3\right)}{\left(4\,\mu 2 - 1 + 4\,\mu 2^2\right)^2} \quad \textbf{(2.3)}$$

```
> op(3,numer(EQmu2)) ; SOLmu2 := solve( %, {mu2} ) ; evalf([%]) ;
```

$$4\,\mu 2^2 + 8\,\mu 2 - 3$$

$$SOLmu2 := \left\{\mu 2 = -1 + \frac{1}{2}\sqrt{7}\right\},\ \left\{\mu 2 = -1 - \frac{1}{2}\sqrt{7}\right\}$$

$$\left[\{\mu 2 = 0.322875656\},\ \{\mu 2 = -2.322875656\}\right] \quad \textbf{(2.4)}$$

```
> SOL0nz := op(simplify(subs(SOLmu2,SOLXY))),op(SOLmu2[1]) ;
```

$$SOL0nz := x = -\frac{1}{2}\,\frac{\left(-2 + \sqrt{7}\right)^2}{-3 + \sqrt{7}},\ y = -\frac{1}{2}\,\frac{-2 + \sqrt{7}}{-3 + \sqrt{7}},\ \mu 2 = -1 + \frac{1}{2}\sqrt{7} \quad \textbf{(2.5)}$$

Check that the solution satisfy the constraints

```
> simplify(subs( SOL0nz, [g1,g2] )) ; evalf(%) ;
```

$$\left[\frac{-45 + 17\sqrt{7}}{\left(-3 + \sqrt{7}\right)^2},\ 0\right]$$

$$[-0.1771243292,\ 0.] \quad \textbf{(2.6)}$$

NO, the solution is NOT feasible.

Try mu2=0 and mu1 free

```
> EQS00 := subs(mu2=0,[EQ||(1..4)] ) ;
```

$$EQS00 := \left[2\,x - y + 2\,\mu 1\,x,\ -x + 2\,\mu 1\,y,\ \mu 1\left(1 - x^2 - y^2\right),\ 0\right] \quad \textbf{(3.1)}$$

```
> SOLxy := solve( EQS00[1..2], {x,y} ) ;
```

$$SOLxy := \{x = 0,\ y = 0\} \quad \textbf{(3.2)}$$

```
> subs( SOLxy, EQS00[3] ) ;
```

$$\mu 1 \quad \textbf{(3.3)}$$

Then mu1 = 0, already considered

Try mu1 and mu2 free

```
> EQ3/mu1;EQ4/mu2;
```

$$1 - x^2 - y^2$$
$$1 - (x-1)^2 - y^2 \quad (4.1)$$

```
> solve( [EQ3/mu1,EQ4/mu2], {x,y} ) ; SOLxy := allvalues(%) ;
```

$$\left\{x = \frac{1}{2}, y = \frac{1}{2}\,RootOf\left(\_Z^2 - 3\right)\right\}$$
$$SOLxy := \left\{x = \frac{1}{2}, y = \frac{1}{2}\sqrt{3}\right\}, \left\{x = \frac{1}{2}, y = -\frac{1}{2}\sqrt{3}\right\} \quad (4.2)$$

Check if the solution is correct

```
> simplify(subs( SOLxy, [EQ1,EQ2] )); SOLmu12 := op(solve( %, [mu1,
  mu2] )) ; evalf(%) ;
```

$$\left[1 - \frac{1}{2}\sqrt{3} + \mu 1 - \mu 2, -\frac{1}{2} + \mu 1\sqrt{3} + \mu 2\sqrt{3}\right]$$
$$SOLmu12 := \left[\mu 1 = -\frac{1}{2} + \frac{1}{3}\sqrt{3}, \mu 2 = -\frac{1}{6}\sqrt{3} + \frac{1}{2}\right]$$
$$\left[\mu 1 = 0.0773502693, \mu 2 = 0.2113248653\right] \quad (4.3)$$

Check that the solution satisfy the constraints

```
> subs( SOLxy, [g1,g2] ) ;
```

$$[0, 0] \quad (4.4)$$

Ok, it is feasible, now check second order condition. The active constraint is only g2 ==> find the kernel of the gradient of g2

```
> gradg1 := subs(SOLxy,<diff(g1,x)|diff(g1,y)>) ;
  gradg2 := subs(SOLxy,<diff(g2,x)|diff(g2,y)>) ;
```

$$gradg1 := \begin{bmatrix} -1 & -\sqrt{3} \end{bmatrix}$$
$$gradg2 := \begin{bmatrix} 1 & -\sqrt{3} \end{bmatrix} \quad (4.5)$$

```
> K := <gradg1,gradg2> ;
```

$$K := \begin{bmatrix} -1 & -\sqrt{3} \\ 1 & -\sqrt{3} \end{bmatrix} \quad (4.6)$$

Find the null space of gradg2

```
>  N := op(NullSpace(K)) ;
```

$$N := \quad (4.7)$$